Global numbering of nodes (italics), elements, and degrees of freedom (numbers on vectors):



Form element stiffiness matrix from column vectors (see text p. 56):

```
with(linalg); a1:=matrix(4,1,[-c,-s,c,s]);a2:=matrix(1,4,[-c,-s,c,s]);
```

$$a1 := \begin{bmatrix} -c \\ -s \\ c \\ s \end{bmatrix}$$

$$a2 := \begin{bmatrix} -c & -s & c & s \end{bmatrix}$$

Multiply these to get element stiffness matrix of Eq. 2.10:

```
k:=evalm( (A*E/L) * a1 &* a2 );
```

$$k := \begin{bmatrix} \frac{A\,E\,c^2}{L} & \frac{A\,E\,c\,s}{L} & -\frac{A\,E\,c^2}{L} & -\frac{A\,E\,c\,s}{L} \\ \frac{A\,E\,c\,s}{L} & \frac{A\,E\,s^2}{L} & -\frac{A\,E\,c\,s}{L} & -\frac{A\,E\,s^2}{L} \\ -\frac{A\,E\,c^2}{L} & -\frac{A\,E\,c\,s}{L} & \frac{A\,E\,c^2}{L} & \frac{A\,E\,c\,s}{L} \\ -\frac{A\,E\,c\,s}{L} & -\frac{A\,E\,s^2}{L} & \frac{A\,E\,c\,s}{L} & \frac{A\,E\,s^2}{L} \end{bmatrix}$$

Trigonometric relations

```
c:=cos(theta);s:=sin(theta); theta:=arctan((y[2]-y[1])/(x[2]-x[1]));
```

$$c := \cos(\theta)$$

$$s := \sin(\theta)$$

$$\theta := \arctan\left(\frac{y_2 - y_1}{x_2 - x_1}\right)$$

Nodal coordinates of element 1

```
x[1]:=0;y[1]:=0;x[2]:=1.5;y[2]:=.25;
```

Set precision, get length

```
Digits:=4;L:=sqrt((x[2]-x[1])^2 + (y[2]-y[1])^2 );
```

$$L := 1.521$$

Define area and modulus (unprotecting E this way is dangerous)

```
A:=3.142e-4;unprotect(E);E:=210e9;
```

$$A := .0003142$$

$$E := .210\ 10^{12}$$

Evaluate stiffness matrix, save as k1

```
k1:=map(eval,k);
```

$$k1 := \begin{bmatrix} .4221\ 10^8 & .7035\ 10^7 & -.4221\ 10^8 & -.7035\ 10^7 \\ .7035\ 10^7 & .1173\ 10^7 & -.7035\ 10^7 & -.1173\ 10^7 \\ -.4221\ 10^8 & -.7035\ 10^7 & .4221\ 10^8 & .7035\ 10^7 \\ -.7035\ 10^7 & -.1173\ 10^7 & .7035\ 10^7 & .1173\ 10^7 \end{bmatrix}$$

Redefine nodal coordinates, for element 2

```
x[1]:=1.5;y[1]:=.25;x[2]:=0;y[2]:=.5;
```

Reevaluate stiffness matrix, save as k2

```
k2:=map(eval,k);
```

$$k2 := \begin{bmatrix} .4221\ 10^8 & -.7035\ 10^7 & -.4221\ 10^8 & .7035\ 10^7 \\ -.7035\ 10^7 & .1173\ 10^7 & .7035\ 10^7 & -.1173\ 10^7 \\ -.4221\ 10^8 & .7035\ 10^7 & .4221\ 10^8 & -.7035\ 10^7 \\ .7035\ 10^7 & -.1173\ 10^7 & -.7035\ 10^7 & .1173\ 10^7 \end{bmatrix}$$

Define global stiffness matrix

```
K:=matrix( 6,6,[   [ k1[1,1], k1[1,2], k1[1,3], k1[1,4], 0,  0 ],
                   [ k1[2,1], k1[2,2], k1[2,3], k1[2,4], 0,  0 ],
                   [ k1[3,1], k1[3,2], k1[3,3]+k2[1,1], k1[3,4]+k2[1,2], k2[1,3], k2[1,4] ],
                   [ k1[4,1], k1[4,2], k1[4,3]+k2[2,1], k1[4,4]+k2[2,2], k2[2,3], k2[2,4] ],
                   [ 0, 0,   k2[3,1],  k2[3,2], k2[3,3], k2[3,4] ],
                   [ 0, 0,   k2[4,1],  k2[4,2], k2[4,3], k2[4,4] ] ] );
```

$$K := \begin{bmatrix} .4221\ 10^8, & .7035\ 10^7, & -.4221\ 10^8, & -.7035\ 10^7, & 0, & 0 \\ .7035\ 10^7, & .1173\ 10^7, & -.7035\ 10^7, & -.1173\ 10^7, & 0, & 0 \\ -.4221\ 10^8, & -.7035\ 10^7, & .8442\ 10^8, & 0, & -.4221\ 10^8, & .7035\ 10^7 \\ -.7035\ 10^7, & -.1173\ 10^7, & 0, & .2346\ 10^7, & .7035\ 10^7, & -.1173\ 10^7 \\ 0, & 0, & -.4221\ 10^8, & .7035\ 10^7, & .4221\ 10^8, & -.7035\ 10^7 \\ 0, & 0, & .7035\ 10^7, & -.1173\ 10^7, & -.7035\ 10^7, & .1173\ 10^7 \end{bmatrix}$$

Solve for unknown displacements - expand rows 3 and 4 of system (dof 1,2,5,6 known to have zero displacement):

```
row3:= K[3,3]*u[3] + K[3,4]*u[4] = 0; row4:= K[4,3]*u[3] + K[4,4]*u[4] = -2000;
```

$$row3 := .8442\ 10^8\ u_3 = 0$$

$$row4 := .2346\ 10^7\ u_4 = -2000$$

Solve for unknown displacements:

```
solve({row3,row4},{u[3],u[4]});
```

$$\left\{ u_3 = 0,\ u_4 = -.0008526 \right\}$$

Solve for unknown forces from **f=KU:**
Left-hand side (displacement) vector:

```
U:=matrix(6,1,[0,0,0,-.0008526,0,0]);
```

$$U := \begin{bmatrix} 0 \\ 0 \\ 0 \\ -.0008526 \\ 0 \\ 0 \end{bmatrix}$$

Form **KU** product to get right-hand side (force) vector

```
'f'=evalm(K&*U);
```

$$f = \begin{bmatrix} 5998. \\ 1000. \\ 0 \\ -2000. \\ -5998. \\ 1000. \end{bmatrix}$$